// 编译：240884432
// 简称: vip13
// 名称: vip13
// 类别: 公式应用
// 类型: 用户应用
// 输出: Void
//------------------------------------------------------------------------
Params

	Numeric X(45);
	Numeric BOLL_N(20);		//周期	
	Numeric Offset(2);		//标准差倍数  
	//跟踪出场参数
	Numeric TrailingStopRate(30); //跟踪止损比例.	
	Numeric Fund(20000); //投入保证金；
Vars
	Numeric Lots(0);
	Series<Numeric> UpLine;		//上轨
	Series<Numeric> DownLine;		//下轨 
	Series<Numeric> MidLine;	//中间线
	Series<Numeric> Band;
	Numeric RSI_N(14);
	Numeric OverSold(30) ; //超卖
	Numeric OverBought(70) ; //超买
	//出场变量
	Series<Numeric> bar_entry_count;
	Series<Numeric> NetChgAvg( 0 );
	Series<Numeric> TotChgAvg( 0 );
	Numeric SF( 0 );
	Numeric Change( 0 );	
	Numeric ChgRatio( 0 ) ;
	Numeric RSIValue;
	Series<Bool> kaiduo;
	Series<Bool> kaikong;
	Global Integer KG;
	Series<Numeric> HH;
	Series<Numeric> LL;
	Series<Numeric> HD;
	Series<Numeric> LD;
	Series<Numeric> entBar;
	Series<Numeric> entPrice ;
	//跟随出场
	Series<Numeric> HighestLowAfterEntry;	//开仓后出现的最高的最低价
	Series<Numeric> LowestHighAfterEntry;	//开仓后出现的最低的最高价	
	Series<Numeric> liQKA(1);
	Series<Numeric> DliqPoint;
	Series<Numeric> KliqPoint;
	Plot plt1;
		//Bar更新事件函数，参数indexs表示变化的数据源图层ID248844320数组
		//Bar更新事件函数，参数indexs表示变化的数据源图层ID121233数组
Defs
	//跟随出场：计算入场后的最高的最低价、最低的最高价, 及对应的止损线
	Integer Trailing_Stop(Numeric Enable_Display = 1)
	{
		
		If(MarketPosition > 0)
		{
			If(CurrentBar == entBar)
			{
				HighestLowAfterEntry	= L ;		
			}Else
			{
				HighestLowAfterEntry = MAX(HighestLowAfterEntry, L);				
			}
		}Else
		If(MarketPosition < 0)
		{	
			If(CurrentBar == entBar)
			{
				LowestHighAfterEntry = H ;		
			}Else
			{
				LowestHighAfterEntry = MIN(LowestHighAfterEntry, H);				
			}
		}
			Commentary("LowestHighAfterEntry"+text(LowestHighAfterEntry));
			Commentary("HighestLowAfterEntry"+text(HighestLowAfterEntry));
		
		/*If(MarketPosition <> 0 And CurrentBar > entBar)
		{
			liQKA = Max(liQKA - 0.1, 0.5);
		}Else
		{
			liQKA=1;
		}*/
		
		DliqPoint = HighestLowAfterEntry - (Open * TrailingStopRate /1000) * liQKA ;
		KliqPoint = LowestHighAfterEntry + (Open * TrailingStopRate /1000) * liQKA ;
		
		If(Enable_Display ==1)
		{
			If(MarketPosition >0)
				PlotAuto("DliqPoint", DliqPoint, DliqPoint, Green, Enum_Cross);
			If(MarketPosition <0)
				PlotAuto("KliqPoint", KliqPoint, KliqPoint, Magenta, Enum_Cross);
		}
		
		Return 1;
	}
	//开仓时 初始化变量
	Numeric Initialize_Vars_On_Entry()
	{
		entBar = CurrentBar;
		entPrice = EntryPrice;
		liQKA = 1;
		
		Return 1;
	}
	//平仓时 重置变量
	Numeric Reset_Vars_On_Exit()
	{
		entBar = 0;
		entPrice = 0;
		
		Return 1;
	}
	//布林通道计算
	Numeric BoLL()
	{
		
		MidLine = AverageFC(Close,BOLL_N);
		Band = StandardDev(Close,BOLL_N,2); 
		UpLine = MidLine + Offset * Band;
		DownLine = MidLine - Offset * Band; 
		Return 1;
	}
	//RSI指标计算
	Numeric RSI()
	{
		If(CurrentBar <= RSI_N - 1)
		{
			NetChgAvg = ( Close - Close[RSI_N] ) / RSI_N ;
			TotChgAvg = Average( Abs( Close - Close[1] ), RSI_N ) ;
		}Else
		{
			SF = 1/RSI_N;
			Change = Close - Close[1] ;
			NetChgAvg = NetChgAvg[1] + SF * ( Change - NetChgAvg[1] ) ;
			TotChgAvg = TotChgAvg[1] + SF * ( Abs( Change ) - TotChgAvg[1] ) ;	
		}

		If( TotChgAvg <> 0 )
		{
		ChgRatio = NetChgAvg / TotChgAvg;
		}else
		{
		ChgRatio = 0 ;
		}	
		RSIValue = 50 * ( ChgRatio + 1 );	
		
		Return 1;
	}	
	
Events
	//此处实现事件函数
	
	//初始化事件函数，策略运行期间，首先运行且只有一次，应用在订阅数据等操作
	OnInit()
	{
		//与数据源有关
		Range[0:DataCount-1]
		{
			//=========数据源相关设置==============
			AddDataFlag(Enum_Data_RolloverBackWard());	//设置后复权

			AddDataFlag(Enum_Data_RolloverRealPrice());	//设置映射真实价格

			AddDataFlag(Enum_Data_AutoSwapPosition());	//设置自动换仓

			//AddDataFlag(Enum_Data_IgnoreSwapSignalCalc());	//设置忽略换仓信号计算
			plt1.figure(0); 
		}
		

	}

	//Bar更新事件函数，参数indexs表示变化的数据源图层ID数组
	OnBar(ArrayRef<Integer> indexs)
	{
		
		plt1.line("资金曲线",Portfolio_CurrentEquity()); //画线
		//计算手数，FUND是投入的保证金，最低下单一手
		Lots = Max(1, Round(Fund/(O*ContractUnit*BigPointValue* MarginRatio/rollover), 0));
		//计算BOLL
		BoLL();
		//计算RSI
		RSI();
		//计算顶部背离
		kaikong=RSIValue>70 and CrossUnder(C,UpLine);
		//计算底部背离
		kaiduo=RSIValue<30 and CrossOver(C,DownLine);
		
		//背离后的上轨道
		HH=Highest(H,X);
		//背离后的下轨道
		LL=Lowest(L,X);
		
		//满足背离条件时
		if( kaiduo[1])
		{
			KG=1;
			HD=HH;
			
		}
		if( kaikong[1])
		{
			KG=-1;
			LD=LL;
		}

		if(MarketPosition!=1 and KG==1 and H>=HD[1])
		{
			Buy(Lots,max(open,HD[1])); 
			bar_entry_count=0;  
			//保存开仓价格
			HighestLowAfterEntry = max(open,HD[1]); 
			//变量初始赋值
			Initialize_Vars_On_Entry();
			
		}
		if(MarketPosition!=-1 and KG==-1 and L<=LD[1])
		{
			SellShort(Lots,min(open,LD[1]));
			bar_entry_count=0;  
			//保存开仓价格
			LowestHighAfterEntry = min(open,LD[1]);
			//变量初始赋值
			Initialize_Vars_On_Entry();
		}
		
		/*计算跟踪止损线*/
		Trailing_Stop(1); 	//参数 == 1 时，在主图绘制止损线
		
		/*跟踪止损*/
		If(MarketPosition > 0 And L <= DliqPoint[1] And CurrentBar > entBar)//多头跟踪止损
		{
			Sell(0, Min(Open, DliqPoint[1]));
			KG=0;
			Commentary("多头跟踪止损");
			//变量重置
			Reset_Vars_On_Exit();
		}
		
		If(MarketPosition < 0 And H >= KliqPoint[1] And CurrentBar > entBar)//空头跟踪止损
		{
			BuyToCover(0, Max(Open, KliqPoint[1]));
			KG=0;
			Commentary("空头跟踪止损");
			//变量重置
			Reset_Vars_On_Exit();
		}
		
		if(MarketPosition<>0)
		{
			bar_entry_count=bar_entry_count+1;

	}

